
1.  Introduction

   The iCalendar [RFC5545] data format is used to represent calendar
   data and is used with iTIP [RFC5546] to handle scheduling operations
   between calendar users. iCalendar is in widespread use, and in
   accordance with provisions in that specification, extension elements
   have been added by various vendors to the data format in order to
   support and enhance capabilities.  This specification collects a
   number of these ad-hoc extensions and uses the new IANA registry
   capability defined in [RFC5545] to register standard variants with
   clearly defined definitions and semantics.  In addition, some new
   elements are introduced for features that vendors have recently been
   requesting.

2.  Conventions Used in This Document

   The key words "MUST", "MUST NOT", "REQUIRED", "SHALL", "SHALL NOT",
   "SHOULD", "SHOULD NOT", "RECOMMENDED", "NOT RECOMMENDED", "MAY", and
   "OPTIONAL" in this document are to be interpreted as described in
   [RFC2119].

   The notation used in this memo is the ABNF notation of [RFC5234] as
   used by iCalendar [RFC5545].  Any syntax elements shown below that
   are not explicitly defined in this specification come from iCalendar
   [RFC5545].

3.  Backwards Compatible Extension Properties

   iCalendar defines properties which can have different value types
   indicated by a "VALUE" parameter.  The definition of a property
   specifies a "default" value type that is assumed to be used when no
   "VALUE" parameter is present.  However, this poses a problem to
   iCalendar parser/generator software that does not know about the
   default values for new properties.  For example, if a new property
   "FOO" were defined with a default value type of URI, and a URI value
   with a comma was used, an iCalendar generator not aware of this fact
   would likely treat the property value as "TEXT" and apply backslash
   escaping to the comma in the value, effectively making it an invalid
   URI value.

   To avoid this problem, this specification recommends that all
   properties not defined in [RFC5545], always include a "VALUE"
   parameter, if the type is other than "TEXT". i.e., in the example
   above, the "FOO" property would have a "VALUE=URI" parameter.  This
   allows iCalendar parser/generator software to track the correct types
   of unknown properties.

https://datatracker.ietf.org/doc/html/rfc5545
https://datatracker.ietf.org/doc/html/rfc5546
https://datatracker.ietf.org/doc/html/rfc5545
https://datatracker.ietf.org/doc/html/rfc2119
https://datatracker.ietf.org/doc/html/rfc5234
https://datatracker.ietf.org/doc/html/rfc5545
https://datatracker.ietf.org/doc/html/rfc5545
https://datatracker.ietf.org/doc/html/rfc5545

   New properties defined in this specification use the term "no
   default" in the "Value Type" definition to indicate that the "VALUE"
   parameter has to be included.

4.  Modifications to Calendar Components

   The following changes to the syntax defined in iCalendar [RFC5545]
   are made here.  New elements are defined in subsequent sections.

   calprops /= *(
                ;
                ; The following are OPTIONAL,
                ; but MUST NOT occur more than once.
                ;
                uid / url / timezid /
                refresh / valid / color /
                ;
                ; The following are OPTIONAL,
                ; and MAY occur more than once.
                ;
                name / description / image
                ;
                )

   eventprop /= *(
                 ;
                 ; The following are OPTIONAL,
                 ; but MUST NOT occur more than once.
                 ;
                 color /
                 ;
                 ; The following are OPTIONAL,
                 ; and MAY occur more than once.
                 ;
                 image
                 ;
                 )

   todoprop /= *(
                ;
                ; The following are OPTIONAL,
                ; but MUST NOT occur more than once.
                ;
                color /
                ;
                ; The following are OPTIONAL,

https://datatracker.ietf.org/doc/html/rfc5545

                ; and MAY occur more than once.
                ;
                image
                ;
                )

   jourprop /= *(
                ;
                ; The following are OPTIONAL,
                ; but MUST NOT occur more than once.
                ;
                color /
                ;
                ; The following are OPTIONAL,
                ; and MAY occur more than once.
                ;
                image
                ;
                )

5.  Properties

5.1.  NAME Property

   Property Name:  NAME

   Purpose:  This property specifies the name of the calendar.

   Value Type:  TEXT

   Property Parameters:  IANA, non-standard, alternate text
      representation, and language property parameters can be specified
      on this property.

   Conformance:  This property can be specified multiple times in an
      iCalendar object.  However, each property MUST represent the name
      of the calendar in a different language.

   Description:  This property is used to specify a name (a short, one-
      line description) of the iCalendar object that can be used by
      calendar user agents when presenting the calendar data to a user.
      Whilst a calendar only has a single name, multiple language
      variants can be specified by including this property multiple
      times with different "LANGUAGE" parameter values on each.

   Format Definition:  This property is defined by the following
      notation:

   name       = "NAME" nameparam ":" text CRLF

   nameparam  = *(
                 ;
                 ; The following are OPTIONAL,
                 ; but MUST NOT occur more than once.
                 ;
                 (";" altrepparam) / (";" languageparam) /
                 ;
                 ; The following is OPTIONAL,
                 ; and MAY occur more than once.
                 ;
                 (";" other-param)
                 ;
                 )

   Example:  The following is an example of this property:

   NAME:Company Vacation Days

5.2.  DESCRIPTION Property

   This specification modifies the definition of the "DESCRIPTION"
   property to allow it to be defined on an iCalendar object.  The
   following additions are made to the definition of this property.

   Purpose:  This property specifies the description of the calendar.

   Conformance:  This property can be specified multiple times in an
      iCalendar object.  However, each property MUST represent the
      description of the calendar in a different language.

   Description:  This property is used to specify a lengthy textual
      description of the iCalendar object that can be used by calendar
      user agents when describing the nature of the calendar data to a
      user.  Whilst a calendar only has a single description, multiple
      language variants can be specified by including this property
      multiple times with different "LANGUAGE" parameter values on each.

5.3.  UID Property

   This specification modifies the definition of the "UID" property to
   allow it to be defined on an iCalendar object.  The following
   additions are made to the definition of this property.

   Purpose:  This property specifies the persistent, globally unique
      identifier for the iCalendar object.

   Conformance:  This property can be specified once in an iCalendar
      object.

5.4.  URL Property

   This specification modifies the definition of the "URL" property to
   allow it to be defined on an iCalendar object.  The following
   additions are made to the definition of this property.

   Purpose:  This property specifies a URL from where the calendar data
      was retrieved or where it can be refreshed.

   Conformance:  This property can be specified once in an iCalendar
      object.

   Description:  This property specifies a URL identifying the source of
      the calendar data and a location from where updates can be
      retrieved.

5.5.  TIMEZONE-ID Property

   Property Name:  TIMEZONE-ID

   Purpose:  This property specifies the default time zone identifier
      for the iCalendar object as a whole.

   Value Type:  TEXT

   Property Parameters:  IANA and non-standard property parameters can
      be specified on this property.

   Conformance:  This property can be specified once in an iCalendar
      object.

   Description:  This property specifies a time zone identifier that
      represents the default timezone for which floating time or all-day
      events in the iCalendar object can be assumed to be relative to.
      It can also be used to choose an initial time zone for use when
      creating new components in the iCalendar object.  A "VTIMEZONE"
      component having a "TZID" property matching the value specified in
      this property MUST be present in the iCalendar object.

   Format Definition:  This property is defined by the following
      notation:

   timezid      = "TIMEZONE-ID" timezidparam
                    ":" text CRLF
                    ;Same value syntax as "TZID" property.

   timezidparam = *(";" other-param)

   Example:  The following is an example of this property:

   TIMEZONE-ID:America/New_York

5.6.  REFRESH-INTERVAL Property

   Property Name:  REFRESH-INTERVAL

   Purpose:  This property specifies a suggested minimum interval for
      polling for changes of the calendar data from the original source
      of that data.

   Value Type:  DURATION - no default

   Property Parameters:  IANA and non-standard property parameters can
      be specified on this property.

   Conformance:  This property can be specified once in an iCalendar
      object.

   Description:  This property specifies a positive duration that gives
      a suggested minimum polling interval for checking for updates to
      the calendar data.  The value of this property SHOULD be used by
      calendar user agents to limit the polling interval for calendar
      data updates to the minimum interval specified.

   Format Definition:  This property is defined by the following
      notation:

   refresh      = "REFRESH-INTERVAL" refreshparam
                    ":" dur-value CRLF
                    ;consisting of a positive duration of time.

   refreshparam = *(
                   ;
                   ; The following is REQUIRED,
                   ; but MUST NOT occur more than once.
                   ;
                   (";" "VALUE" "=" "DURATION") /
                   ;
                   ; The following is OPTIONAL,
                   ; and MAY occur more than once.
                   ;
                   (";" other-param)
                   ;
                   )

   Example:  The following is an example of this property:

   REFRESH-INTERVAL;VALUE=DURATION:P1W

5.7.  VALID Property

   Property Name:  VALID

   Purpose:  This property specifies when the calendar data is valid.

   Value Type:  DATE-TIME or PERIOD

   Property Parameters:  IANA and non-standard property parameters can
      be specified on this property.

   Conformance:  This property can be specified once in an iCalendar
      object.

   Description:  This property specifies a time period for which the
      calendar data can assumed to be valid.  If a "PERIOD" value type
      is used, the validity is assumed to be a time range defined by the
      start and end of the period.  If a "DATE-TIME" value is used, the
      date indicates when end date when the calendar data is no longer
      valid.  Once the end date has been reached (or at some convenient
      time prior) clients SHOULD refresh the calendar data to determine
      whether an update is available, extending the range of validity.
      The value MUST be specified in UTC.

   Format Definition:  This property is defined by the following
      notation:

   valid      = "VALID" validparam ":" validval CRLF

   validparam   = *(
                   ;
                   ; The following is REQUIRED,
                   ; but MUST NOT occur more than once.
                   ;
                   (";" "VALUE" "=" ("DATE-TIME" / "PERIOD")) /
                   ;
                   ; The following is OPTIONAL,
                   ; and MAY occur more than once.
                   ;
                   (";" other-param)
                   ;
                   )

   validval     = date-time / period
   ;Value MUST match value type. Value MUST be in UTC.

   Example:  The following is an example of this property:

   VALID;VALUE=DATE-TIME:20120609T000000Z
   VALID;VALUE=PERIOD:20120609T000000Z/P365D

5.8.  COLOR Property

   Property Name:  COLOR

   Purpose:  This property specifies a color used for displaying the
      calendar, event, todo, or journal data.

   Value Type:  TEXT.  The value MUST be three COLON-separated INTEGER
      values.

   Property Parameters:  IANA and non-standard property parameters can
      be specified on this property.

   Conformance:  This property can be specified once in an iCalendar
      object, or "VEVENT", "VTODO", or "VJOURNAL" calendar components.

   Description:  This property specifies a color that client MAY use
      when presenting the relevant data to a user.  Typically this would
      appear as the "background" color of events or tasks.  The value
      MUST be an RGB value with integer value components in the range
      0..255.  If a color is specified on a "VEVENT", "VTODO" or

      "VJOURNAL" that SHOULD override any color specified on the
      enclosing iCalendar object.

   Format Definition:  This property is defined by the following
      notation:

   color          = "COLOR" colorparam ":" colorvalue CRLF

   colorparam     = *(";" other-param)

   colorvalue     = integer ":" integer ":" integer
                    ; Red, green, and blue values in the range
                    ; 0 - 255.

   Example:  The following is an example of this property:

   COLOR:255:0:255

5.9.  IMAGE Property

   Property Name:  IMAGE

   Purpose:  This property specifies an image associated with the
      calendar or a calendar component.

   Value Type:  URI or BINARY - no default.  The value MUST refer to or
      be data with a media type of "image".

   Property Parameters:  IANA, non-standard, display, inline encoding,
      and value data type property parameters can be specified on this
      property.  The format type parameter can be specified on this
      property and is RECOMMENDED for inline binary encoded content
      information.

   Conformance:  This property can be specified multiple times in an
      iCalendar object, or "VEVENT", "VTODO", or "VJOURNAL" calendar
      components.

   Description:  This property specifies an image for an iCalendar
      object or a calendar component via a uri or directly with inline
      data that can be used by calendar user agents when presenting the
      calendar data to a user.  Multiple properties MAY be used to
      specify alternative sets of images with, for example, varying
      media subtypes, resolutions or sizes.  When multiple properties
      are present, calendar user agents SHOULD display only one of them,
      picking one that provides the most appropriate image quality, or
      display none.  The "DISPLAY" parameter is used to indicate the
      intended display mode for the image.  An "ALTURI" parameter is

      used to provide a "clickable" image where the URI in the parameter
      value can be "launched" by a click on the image in the calendar
      user agent.

   Format Definition:  This property is defined by the following
      notation:

   image      = "IMAGE" imageparam ( ":" uri ) /
                (
                  ";" "ENCODING" "=" "BASE64"
                  ";" "VALUE" "=" "BINARY"
                  ":" binary
                )
                CRLF

   imageparam = *(
                 ;
                 ; The following is REQUIRED,
                 ; but MUST NOT occur more than once.
                 ;
                 (";" "VALUE" "=" "URI") /
                 ;
                 ; The following is OPTIONAL for a URI value,
                 ; RECOMMENDED for a BINARY value,
                 ; and MUST NOT occur more than once.
                 ;
                 (";" fmttypeparam) /
                 ;
                 ; The following are OPTIONAL,
                 ; and MUST NOT occur more than once.
                 ;
                 (";" alturiparam) / (";" displayparam) /
                 ;
                 ; The following is OPTIONAL,
                 ; and MAY occur more than once.
                 ;
                 (";" other-param)
                 ;
                 )

   Example:  The following is an example of this property:

   IMAGE;VALUE=URI;DISPLAY=BACKGROUND;FMTTYPE=image/png:h
    ttp://example.com/images/party.png



Daboo                    Expires August 4, 2013                [Page 12]



Internet-Draft        iCalendar Property Extensions         January 2013

6.  Property Parameters

6.1.  ALTURI Property Parameter

   Parameter Name:  ALTURI

   Purpose:  To specify a URI alternative to a property value.

   Format Definition:  This property parameter is defined by the
      following notation:

   alturiparam = "ALTURI" "=" DQUOTE uri DQUOTE

   Description:  This property parameter MAY be specified on "IMAGE"
      properties.

   Example:

   IMAGE;VALUE=URI;FMTTYPE=image/png:ALTURI="http://example.co
    m/clicked-image1":http://example.com/images/party.png

6.2.  DISPLAY Property Parameter

   Parameter Name:  DISPLAY

   Purpose:  To specify different ways in which an image for a calendar
      or component can be displayed.

   Format Definition:  This property parameter is defined by the
      following notation:

   displayparam = "DISPLAY" "="
                  ("BADGE" /       ; A small "badge" image
                   "BACKGROUND" /  ; Use as a background image
                   "OVERLAY" /     ; Use as an overlay image
                   "BANNER" /      ; Use as a "banner" across the top
                   x-name /        ; Experimental type
                   iana-token)     ; Other IANA registered type
                                   ;
                                   ; Default is BADGE

   Description:  This property parameter MAY be specified on "IMAGE" or
      "IMAGE" properties.  In the absence of this parameter, the value
      "BADGE" MUST be used for the default behavior.  The value
      determines how a client ought to present an image supplied in
      iCalendar data to the user.

      Values for this parameter are registered with IANA as per
Section 8.3.  New values can be added to this registry following

      the procedure outlined in Section 8.2.1 of [RFC5545].

      Servers and clients MUST handle x-name and iana-token values they
      don't recognize by not displaying any image at all.

   Example:

   IMAGE;VALUE=URI;DISPLAY=BANNER;FMTTYPE=image/png:http://exa
    mple.com/images/weather-cloudy.png

7.  Security Considerations

   Several of the new properties or parameters defined by this
   specification allow reference to "external" URIs.  Care MUST be taken
   when accessing data at external URIs as malicious content could be
   present.  In addition, access to those URIs could be tracked, leading
   to loss of privacy.

8.  IANA Considerations

8.1.  Property Registrations

   This document defines the following new iCalendar properties to be
   added to the registry defined in Section 8.2.3 of [RFC5545]:

   +------------------+---------+--------------------------------------+
   | Property         | Status  | Reference                            |
   +------------------+---------+--------------------------------------+
   | NAME             | Current | RFCXXXX, Section 5.1                 |
   | DESCRIPTION      | Current | RFC5545 Section 3.8.1.5, RFCXXXX,    |
   |                  |         | Section 5.2                          |
   | UID              | Current | RFC5545 Section 3.8.4.7, RFCXXXX,    |
   |                  |         | Section 5.3                          |
   | URL              | Current | RFC5545 Section 3.8.4.6, RFCXXXX,    |
   |                  |         | Section 5.4                          |
   | TIMEZONE-ID      | Current | RFCXXXX, Section 5.5                 |
   | REFRESH-INTERVAL | Current | RFCXXXX, Section 5.6                 |
   | VALID            | Current | RFCXXXX, Section 5.7                 |
   | COLOR            | Current | RFCXXXX, Section 5.8                 |
   | IMAGE            | Current | RFCXXXX, Section 5.9                 |
   +------------------+---------+--------------------------------------+

https://datatracker.ietf.org/doc/html/rfc5545#section-8.2.1
https://datatracker.ietf.org/doc/html/rfc5545#section-8.2.3
https://datatracker.ietf.org/doc/html/rfc5545#section-3.8.1.5
https://datatracker.ietf.org/doc/html/rfc5545#section-3.8.4.7
https://datatracker.ietf.org/doc/html/rfc5545#section-3.8.4.6

8.2.  Parameter Registrations

   This document defines the following new iCalendar property parameters
   to be added to the registry defined in Section 8.2.4 of [RFC5545]:

          +--------------------+---------+----------------------+
          | Property Parameter | Status  | Reference            |
          +--------------------+---------+----------------------+
          | ALTURI             | Current | RFCXXXX, Section 6.1 |
          | DISPLAY            | Current | RFCXXXX, Section 6.2 |
          +--------------------+---------+----------------------+

8.3.  Display Types Registry

   This document defines the following new iCalendar value registry as
   per Section 8.2.6 of [RFC5545]:

             +--------------+---------+----------------------+
             | Display Type | Status  | Reference            |
             +--------------+---------+----------------------+
             | BADGE        | Current | RFCXXXX, Section 6.2 |
             | BACKGROUND   | Current | RFCXXXX, Section 6.2 |
             | OVERLAY      | Current | RFCXXXX, Section 6.2 |
             | BANNER       | Current | RFCXXXX, Section 6.2 |
             +--------------+---------+----------------------+

Appendix A.  Change History (To be removed by RFC Editor before
             publication)

   Changes in -06:

   1.  Removed BROADCAST/CONFERENCE properties and related parameters.

   Changes in -05:

   1.  Added section with recommendation on handling extension
       properties.

   2.  Added VALID property.

   Changes in -04:

   1.  TZID changed to new property TIMEZONE-ID.

   2.  Minor formal syntax changes.

   Changes in -03:

   1.  Dropped CALENDAR- prefix

   2.  DESCRIPTION, UID and TZID now based on existing RFC5545
       properties

   3.  COLOR now on both the calendar and component level

   4.  IMAGE now on both the calendar and component level

   5.  Added FEATURE and REGION parameters to CONFERENCE property

   6.  Added ALTURI parameter to IMAGE property

   7.  Added FEED value to FEATURE parameter

   8.  Added BROADCAST property and clarified that CONFERENCE is for bi-
       direction channels and BROADCAST is for uni-directional.

   Changes in -02:

https://datatracker.ietf.org/doc/html/rfc5546
https://datatracker.ietf.org/doc/html/rfc5545

   1.  Minor wording changes.

   2.  Interval is now described as the "minimum interval".

   3.  Added CONFERENCE property and INFO parameter.

   Changes in -01:

   1.  Fixed DISPLAY parameter handling of x- and iana tokens to state
       that clients ignore the image if the token is not recognized.

   2.  Allow language variants for CALENDAR-NAME and CALENDAR-
       DESCRIPTION.

   3.  Added registry for DISPLAY values.

